

Komunikasi Antar Pemakai CDS/ISIS

Agus Permadi dan Eko S. Priyocahyono

BAHASA FORMAT CDS/ISIS

Sari: Artikel ini membahas bahasa yang digunakan untuk menulis format tampilan/cetak cantuman CDS/ISIS, khususnya perintah indensi, perintah spasi, perintah imbuhan, dan perintah model luaran.

Pada umumnya data disimpan dalam komputer dengan tujuan agar dapat ditemukan kembali dengan mudah. Kata 'ditemukan' dapat berarti diperlihatkan pada layar komputer atau dicetak di atas kertas. Data yang tampak pada layar komputer atau yang tercetak di atas kertas disebut **luaran**. Bentuk susunan data dalam daftar tersebut disebut **format**. Untuk menciptakan format tertentu, kita harus memberikan perintah kepada komputer untuk mengambil data tertentu dari pangkalan data dan menempatkan data ini menurut susunan atau format tertentu. Agar komputer — dalam hal ini adalah CDS/ISIS — dapat menjalankan perintah tadi, perintah harus diberikan dalam bahasa yang dipahami oleh CDS/ISIS.

Komunikasi manusia dengan komputer sedikit berbeda dari komunikasi antar manusia. Jika seseorang (misalnya A) ingin berbicara dengan orang lain (B), A harus mengetahui bahasa yang dimengerti oleh B. Jika hal ini diabaikan, komunikasi tidak akan dapat berjalan dengan baik. Misalnya, jika A memberikan perintah kepada B dalam bahasa yang tidak dipahami oleh B, maka perintah ini tidak akan dapat dijalankan atau dijalankan dengan tidak benar. Kelebihan manusia adalah ia bisa belajar agar dapat memahami lebih dari satu bahasa; oleh karena itu dua orang dapat berkomunikasi dalam dua bahasa atau lebih.

Pada satu waktu, komputer hanya dapat memahami satu bahasa. Misalnya, jika kita sedang mengetik dengan perangkat lunak WordStar, maka menggarisbawahi kata harus dinyatakan dengan menekan secara serentak tombol-tombol tertentu, yaitu **Ctrl, P** dan **S**. Ketiga tombol ini merupakan kalimat perintah menggarisbawahi yang hanya dipahami oleh WordStar. Untuk memberikan perintah yang sama kepada

perangkat lunak lain, kita harus menggunakan **kalimat** atau **bahasa** lain.

Seperti perangkat lunak lain, CDS/ISIS hanya dapat memahami bahasa tertentu. Dengan bahasa ini kita memberikan perintah kepada CDS/ISIS agar mengerjakan atau melakukan sesuatu. Kita dapat memberikan perintah kepada CDS/ISIS untuk mencetak atau memperlihatkan ruas, misalnya mencetak semua nilai ruas judul yang terdapat dalam pangkalan data. CDS/ISIS juga dapat diminta untuk mencetak ruas dengan mempertimbangkan syarat tertentu, misalnya mencetak semua nilai ruas judul dari buku-buku berbahasa Inggris saja. Cara menulis kalimat perintah yang dapat dipahami oleh CDS/ISIS akan dijelaskan dalam artikel ini.

MEMILIH RUAS DAN SUBRUAS

Untuk mengambil nilai atau data dari sebuah ruas, kita harus menuliskan tengara ruas tersebut. Tengara untuk setiap ruas dalam pangkalan data telah ditentukan dalam TDR atau tabel definisi ruas. Dalam bahasa format, setiap tengara harus didahului oleh huruf 'v' kecil atau besar. Di bawah ini adalah contoh-contoh ruas dan nilainya (isinya) serta luaran yang diperoleh dengan berbagai format. Nilai ruas diperlihatkan sesuai dengan formatnya pada waktu diketik dalam lembar kerja ketik.

PENULIS PERSONAL (300): Hatta, Mohammad

Format: **v300**

Luaran: Hatta, Mohammad

SUBYEK (620): BANK DAN PERBANKAN % INVESTASI

Format: **v620**

Luaran: BANK DAN PERBANKANINVESTASI

PENERBIT (400): aJakarta^bGramedia^c1994

Format: **v400**

Luaran: ^aJakarta^bGramedia^c1994

Format: **v400^a**

Luaran: Jakarta

Format: **v400^b**

Luaran: Gramedia

Format: **v400^c**

Luaran: 1994

Format: **v400^***

Luaran: Jakarta

Penanda subruas **^*** dalam format **v400^*** merupakan perintah agar CDS/ISIS mengambil subruas pertama dari suatu ruas. Cara memisahkan subyek pertama dan kedua dalam luaran dari format **v620** dibahas dalam Subseksi **Imbuhan terulang**. Cara menghilangkan penanda subruas (**^a**, **^b**, dan **^c**) dalam luaran dari format **v400** dibahas dalam Seksi **Perintah model luaran**.

MEMILIH SEBAGIAN DARI NILAI RUAS ATAU SUBRUAS

Untuk keperluan tertentu, kadang-kadang kita hanya perlu mengambil **sejumlah** nilai ruas atau **sejumlah** karakter dalam ruas atau subruas dari **posisi** tertentu. Sarana ini digunakan terutama untuk mengambil sebagian nilai ruas atau subruas yang panjangnya tetap, misalnya Ruas **TANGGAL KETIK** atau subruas tahun dalam Ruas **PENERBIT**.

Untuk menentukan posisi dari mana nilai ruas harus diambil, digunakan perintah ***angka**. Dalam CDS/ISIS, posisi awal adalah **nol**. Jika posisi awal tidak ditentukan, CDS/ISIS akan menggunakan posisi awal nilai ruas.

Untuk mengambil sejumlah karakter dari awal ruas atau subruas, digunakan perintah **.angka**. Untuk mengambil sejumlah karakter dari posisi tertentu, digunakan perintah ***angka.angka**. Jika angka panjang tidak ditentukan, CDS/ISIS akan mengambil data mulai dari posisi yang ditentukan dalam ***angka**

hingga akhir ruas atau subruas. Di bawah ini adalah contoh-contoh ruas dan nilainya serta luaran yang dihasilkan dengan berbagai format:

TANGGAL KETIK (22): 1996-03-29

Format: **v22*2**

Luaran: 96-03-29

Format: **v22*2.2**

Luaran: 96

Format: **v22.7**

Luaran: 1996-03

Format: **v22*8, v22*4.4, v22.4**

Luaran: 29-03-1996

PENERBIT (400): ^aJakarta^bGramedia^c1994

Format: **v400^c*2.2**

Luaran: 94

Format: **v400.5**

Luaran: ^aJak

Perhatikan dua contoh terakhir di atas: Jika penanda subruas (**^c**) dipakai, posisi awal (posisi nol) berada pada karakter pertama dari nilai ruas, yaitu pada angka 1 dalam 1994, bukan pada tanda tudung (**^**) dari penanda subruas (**^c**). Jika penanda subruas tidak dipakai, posisi awal berada pada tanda tudung (**^**) dari penanda subruas pertama.

PERINTAH INDENSI

Jika tidak terdapat perintah khusus dalam format, nilai ruas atau data akan tercetak pada satu baris dengan nilai ruas sebelumnya. Jika nilai ruas melampaui panjang baris yang tersedia, sisanya akan menempati baris baru dan tercetak mulai dari margin kiri. Untuk mencetak nilai ruas beberapa spasi sebelah kanan margin kiri, perintah indensi harus digunakan. Perintah indensi dinyatakan dengan kode (**f,c**) di mana:

f adalah angka (nol dan angka positif) yang menyatakan jumlah spasi sebelum tercetak baris **pertama** dari nilai ruas.

c adalah angka (nol dan angka positif) yang menyatakan jumlah spasi sebelum tercetak baris **kedua dan seterusnya** dari nilai ruas.

Jika *f* saja yang diperlukan, maka *c* dapat dihilangkan (tidak harus ditulis). Misalnya, jika baris pertama harus tercetak pada posisi lima spasi dari margin kiri dan baris kedua dan selanjutnya pada margin kiri, maka perintah dapat ditulis sebagai berikut: (5,0) atau (5). Tetapi jika *c* saja yang diperlukan, maka *f* juga harus ditulis dalam format. Perintah indensi ditempatkan dalam format setelah tengara ruas yang harus dicetak menurut indensi tersebut.

Di bawah ini adalah contoh-contoh ruas dan nilainya serta luaran yang diperoleh dalam berbagai format. Dalam contoh ini satu spasi dinyatakan dengan satu titik; dalam luaran yang sebenarnya, titik ini tidak ada.

JUDUL DAN KEPENGARANGAN (200): ^aPeralatan hiburan dan kesenian tradisional Daerah Istimewa Yogyakarta

Format: **v200^a**

Luaran: Peralatan hiburan dan kesenian tradisional Daerah Istimewa Yogyakarta

Format: **v200^a(5,0)** atau **v200^a(5)**

Luaran:Peralatan hiburan dan kesenian tradisional Daerah Istimewa Yogyakarta

Format: **v200^a(3,5)**

Luaran: ...Peralatan hiburan dankesenian tradisionalDaerah IstimewaYogyakarta

Format: **v200^a(0,5)**

Luaran: Peralatan hiburan dankesenian tradisionalDaerah IstimewaYogyakarta

PERINTAH SPASI

Dalam CDS/ISIS dikenal dua jenis spasi: spasi vertikal dan spasi horizontal. Spasi vertikal adalah spasi antar baris. Spasi horizontal adalah spasi antar kata dalam baris. Untuk format CDS/ISIS terdapat lima perintah untuk mengatur spasi vertikal dan horizontal: *Xn*, *Cn*, */*, *#*, dan *%*.

Xn

Kode ini dipakai untuk menciptakan spasi horizontal sebanyak *n*. Jika jumlah spasi yang tersedia lebih kecil

dari pada *n*, maka nilai ruas atau data akan ditempatkan pada baris baru mulai dari margin kiri. Misalnya, jika posisi berikutnya pada suatu baris adalah 78 dan lebar baris adalah 80, maka perintah *x5* akan menempatkan nilai ruas di baris berikutnya pada margin (posisi 1), bukan pada posisi 2.

Perintah *x3* dan *x8* dalam format di bawah ini berturut-turut menciptakan tiga spasi sebelum 'Jakarta' dan delapan spasi sebelum 'Gramedia'.

PENERBIT (400): ^aJakarta^bGramedia^c1994

Format: **x3,v400^a,x8,v400^b**

Luaran: ...Jakarta.....Gramedia

Cn

Kode ini dipakai untuk menempatkan nilai ruas atau data mulai dari posisi atau kolom *n*. Posisi *n* dihitung dari margin kiri, bukan dari posisi yang belum ditempati pada suatu baris. Jika posisi *n* pada suatu baris telah ditempati, maka ruas nilai akan ditempatkan mulai dari kolom *n* pada baris baru.

Perintah *c10* dalam format pertama di bawah ini menempatkan huruf pertama 'G' dari 'Gramedia' pada kolom ke-10, dihitung dari margin kiri atau posisi huruf 'J' dari 'Jakarta'. Pada format kedua, perintah *c10* menempatkan 'Gramedia' pada baris baru mulai dari kolom ke-10 karena kolom ke-10 pada baris pertama telah ditempati oleh huruf 'a' terakhir dari 'Jakarta'. Penggunaan perintah *Xn* pada format ketiga menghindarkan keharusan mengetahui jumlah spasi yang telah ditempati.

PENERBIT (400): ^aJakarta^bGramedia^c1994

Format: **v400^a,c10,v400^b**

Luaran: Jakarta..Gramedia

Format: **c4,v400^a,c10,v400^b**

Luaran: ...Jakarta
.....Gramedia

Format: **c4,v400^a,x3,v400^b**

Luaran: ...Jakarta...Gramedia

/ (garis miring)

Kode ini dipakai untuk menempatkan nilai ruas atau data pada baris baru atau menciptakan spasi vertikal. Dalam format, kode ini ditempatkan sebelum atau mendahului tengara atau data. Perintah */* hampir sama

kegunaannya dengan tombol Enter. Meskipun demikian, sejumlah garis miring belum tentu menciptakan sejumlah spasi vertikal karena perintah / tidak akan dijalankan jika ruas sebelumnya tidak mempunyai nilai. Hal ini akan lebih jelas dengan contoh di bawah ini. Untuk menempatkan data pada baris baru dan juga menciptakan spasi vertikal sebelumnya, pada umumnya digunakan garis miring (/) dan tanda nomor (#).

Catatan: 'Nilai ruas' dalam CDS/ISIS dapat berupa huruf, angka, tanda baca, simbol matematika, atau spasi yang diciptakan dengan menekan tombol Space bar. Karena spasi tidak mudah terlihat dalam ruas yang kosong, pengetik cantuman harus cermat untuk mengetahui apakah suatu ruas benar-benar kosong. Untuk menghapus spasi yang mungkin terdapat dalam suatu ruas yang kosong, tempatkan kursor pada ruas tersebut dan kemudian tekan tombol F2.

Pada luaran dari format pertama di bawah ini, judul dan penerbit tercetak pada baris baru. Pada luaran untuk format kedua, antara penulis dan judul tidak terdapat satu spasi vertikal meskipun terdapat perintah dua garis miring (//). Hal ini menunjukkan bahwa perintah / tidak dimaksudkan untuk menciptakan spasi vertikal. Pada luaran untuk format ketiga, antara judul dan penerbit tidak terdapat spasi vertikal karena Ruas PENULIS PERSONAL (v300) tidak mempunyai nilai.

Cantuman 1:

JUDUL DAN KEPENGARANGAN (200): ^aEko-
nomi pertanian Indonesia
PENULIS PERSONAL (300): Sastraatmaja, Entang
PENERBIT (400): ^aBandung^bAngkasa^c1995

Format: **v300/v200^a(4,0)/c5,v400^b**

Luaran: Sastraatmaja, Entang
....Ekonomi pertanian
Indonesia
....Angkasa

Format: **v300//v200^a(4,0)**

Luaran: Sastraatmaja, Entang
....Ekonomi pertanian
Indonesia

Cantuman 2:

JUDUL DAN KEPENGARANGAN (200): ^aLa-
poran tahunan
PENULIS PERSONAL (300): [kosong]

PENERBIT (400): ^aSamarinda^bDinas Pertanian
^c1996

Format: **v200^a(4,0)/v300/c5,v400^b**

Luaran:Laporan tahunan
....Dinas Pertanian

(nomor)

Kode ini digunakan untuk menempatkan nilai ruas atau data pada baris baru atau menciptakan spasi vertikal. Berbeda dari perintah /, perintah # tetap memindahkan nilai ruas atau data ke baris baru meskipun dalam ruas sebelumnya tidak terdapat nilai. Jika dileakkan setelah /, kode ini dapat mengatasi kekurangan pada perintah /. Di bawah ini adalah contoh gabungan /# dalam format dan luaran yang dihasilkannya. Pada luaran, spasi vertikal ditandai dengan titik-titik; pada luaran yang sebenarnya, titik-titik ini tidak ada.

Cantuman:

JUDUL DAN KEPENGARANGAN (200): ^aLaporan
tahunan

PENULIS PERSONAL (300): [kosong]

PENERBIT (400): ^aSamarinda^bDinas Pertanian
^c1996

Format: **v200^a/#v300/#v400^c**

Luaran: Laporan tahunan [1]
..... [2]
..... [3]
1996 [4]

CDS/ISIS menjalankan format di atas sebagai berikut: Karena Ruas JUDUL DAN KEPENGARANGAN mempunyai nilai, CDS/ISIS siap menempatkan nilai Ruas PENULIS PERSONAL pada baris 3. Karena dalam Ruas PENULIS PERSONAL ternyata tidak terdapat nilai, CDS/ISIS tidak menjalankan perintah / dalam perintah /# yang kedua. CDS/ISIS hanya menjalankan perintah # sebanyak dua kali, yaitu menempatkan nilai Subruas c dari Ruas PENERBIT pada baris baru atau baris 4.

Untuk menghindari spasi vertikal yang timbul seperti pada contoh di atas, perintah % di bawah ini dapat digunakan.

% (tanda persen)

Kode ini digunakan untuk menghapus spasi vertikal yang tidak diperlukan yang timbul jika digu-

nakan perintah / dan dalam suatu ruas tidak terdapat nilai. Di bawah ini adalah contoh format dengan perintah % dan luaran yang dihasilkannya:

Cantuman:

JUDUL DAN KEPENGARANGAN (200): ^aLaporan tahunan

PENULIS PERSONAL (300): [kosong]

PENERBIT (400): ^aSamarinda^bDinas Pertanian ^c1996

Format: c5,v200^a%#v300%#v400^b

Luaran: ...Laporan tahunan [1]
Dinas Pertanian [2]

CDS/ISIS menjalankan perintah di atas sebagai berikut: Karena Ruas JUDUL DAN KEPENGARANGAN mempunyai nilai, maka CDS/ISIS siap menempatkan nilai Ruas PENULIS PERSONAL pada baris 2. Karena dalam Ruas PENULIS PERSONAL ternyata tidak terdapat nilai, maka % (dalam perintah %# yang kedua) membatalkan perintah # dalam perintah %# yang pertama. Kemudian, # (dalam perintah %# yang kedua) menempatkan Subruas b dari Ruas PENERBIT pada baris 2.

IMBUHAN

Imbuan adalah rangkaian karakter (huruf, angka, tanda baca, simbol matematika) atau spasi yang harus tercetak dalam luaran sama seperti yang tertulis dalam format. Imbuan juga dapat berupa kata atau kalimat, misalnya nama ruas. Dalam format, imbuan harus diketik dalam tanda baca tertentu. Dalam format CDS/ISIS terdapat tiga jenis imbuan: **imbuan bersyarat**, **imbuan tak bersyarat**, dan **imbuan terulangkan**.

Imbuan yang ditempatkan dalam format sebelum atau di depan tengara ruas disebut **imbuan awal**. Pada luaran, imbuan awal akan tercetak sebelum atau di depan nilai ruas. Imbuan yang ditempatkan dalam format setelah tengara ruas disebut **imbuan akhir**. Imbuan akhir akan tercetak setelah nilai ruas pada luaran.

Imbuan bersyarat

Dalam format, imbuan bersyarat harus ditempatkan dalam tanda petik (") dan diketik **sebelum** atau **setelah** tengara, misalnya "PENULIS: ". Imbuan

bersyarat akan tercetak jika dalam ruas yang bersangkutan terdapat nilai. Jika digunakan untuk ruas terulangkan (misalnya Ruas PENULIS PERSONAL), imbuan akan tercetak **hanya** sekali. Dalam imbuan bersyarat **tidak** boleh terdapat tanda petik.

Pada luaran dari format 1 di bawah ini, imbuan (PENULIS:) untuk cantuman 2 tidak tercetak karena dalam Ruas PENULIS PERSONAL tidak terdapat nilai. (Perhatikan jumlah spasi antara titik dua dan tanda petik tunggal akhir pada setiap imbuan dalam format. Ini dimaksudkan agar nilai ruas tercetak mulai dari posisi atau kolom yang sama.) Pada luaran untuk format 2, imbuan ((PENERBIT)) ditempatkan setelah tengara ruas dan oleh karena itu tercetak dalam luaran setelah nilai ruas.

Imbuan bersyarat kosong (ditulis sebagai """, tanpa spasi di antara keduanya) dapat digunakan sebagai imbuan awal untuk menciptakan spasi vertikal bersyarat (tercetak jika dalam ruas terdapat nilai). Jika digunakan sebagai imbuan akhir, imbuan bersyarat kosong dapat menghapus titik dan spasi yang tercetak setelah nilai ruas yang disebabkan oleh penggunaan perintah model luaran **mdu**.

Cantuman 1:

JUDUL DAN KEPENGARANGAN (200): ^aMerintis hari esok

PENULIS PERSONAL (300): Intisaruddin

PENERBIT (400): ^aJakarta^bBalai Pustaka^c1989

Cantuman 2:

JUDUL DAN KEPENGARANGAN (200): ^aLaporan tahunan

PENULIS PERSONAL (300): [kosong]

PENERBIT (400): ^aBogor^bPusat Penelitian Tanah ^c1992

Format 1: "PENULIS: ",v300/"JUDUL: ",v200^a/#

Luaran: PENULIS: Intisaruddin
JUDUL: Merintis hari esok

JUDUL: Laporan tahunan

Format 2: v400^b" (PENERBIT)"/#

Luaran: Balai Pustaka (PENERBIT)

Pusat Penelitian Tanah
(PENERBIT)

Imbuhan bersyarat yang ditempatkan di depan tengara ruas tidak mematuhi perintah indensi (f,c) untuk ruas tersebut; oleh karena itu perintah kolom (Cn) harus digunakan jika diperlukan indensi. Hal ini tampak pada luaran dari format pertama (yang menggunakan perintah indensi) dan kedua (menggunakan Cn) di bawah ini.

Cantuman:

JUDUL DAN KEPENGARANGAN (200): ^aPendidikan tinggi dan tenaga kerja tingkat tinggi di Indonesia

Format 1: "JUDUL: ",v200^a(7,7)/#

Luaran: JUDUL: Pendidikan tinggi dan tenaga kerja tingkat tinggi di Indonesia

Format 2: c8,"JUDUL: ",v200^a/#

Luaran:JUDUL: Pendidikan tinggi dan tenaga kerja tingkat tinggi di Indonesia

Imbuhan tak bersyarat

Imbuhan tak bersyarat akan tercetak pada luaran meskipun dalam ruas tidak terdapat nilai. Dalam format, imbuhan tak bersyarat ditempatkan dalam tanda petik tunggal (') dan diketik sebelum atau setelah tengara ruas, misalnya 'JUDUL: '. Pada luaran, imbuhan bersyarat akan terpotong jika tidak dapat ditempatkan dalam satu baris. Untuk menghindarkan pemotongan ini, imbuhan yang terlalu panjang harus ditempatkan dalam dua atau tiga bagian. Dalam imbuhan tidak bersyarat tidak boleh terdapat tanda petik tunggal ('), tetapi tanda petik (") dan garis tegak (|) dapat digunakan.

Untuk format 1 di bawah ini, imbuhan (PENULIS) dalam luaran cantuman 2 tetap tercetak meskipun dalam Ruas PENULIS PERSONAL tidak terdapat nilai. Dalam format 2, imbuhan ditempatkan setelah tengara ruas dan oleh karena itu tercetak setelah nilai ruas.

Cantuman 1:

PENULIS PERSONAL (300): Katoppo, Aristides
JUDUL DAN KEPENGARANGAN (200): ^aSatu abad Kartini

Cantuman 2:

PENULIS PERSONAL (300): [kosong]
JUDUL DAN KEPENGARANGAN (200): ^aLaporan tahunan dan kemajuan

Format 1: 'PENULIS: ',v300/'JUDUL: ',v200^a/#

Luaran: PENULIS: Katoppo, Aristides
JUDUL: Satu abad Kartini

PENULIS:

JUDUL: Laporan tahunan dan kemajuan

Format 2: v300' (PENULIS)'/#

Luaran: Katoppo, Aristides (PENULIS)
(PENULIS)

Imbuhan terulangkan

Imbuhan terulangkan dimaksudkan hanya untuk ruas terulangkan. Pada luaran, imbuhan terulangkan akan tercetak pada setiap nilai ruas. Dalam format, imbuhan terulangkan harus ditempatkan dalam garis tegak (|), misalnya | ; |. Dalam imbuhan terulangkan tidak boleh terdapat garis tegak (|), tetapi tanda petik (") dan petik tunggal (') dapat digunakan. Seperti imbuhan bersyarat, imbuhan terulangkan tercetak jika dalam ruas terdapat nilai.

Imbuhan terulangkan terdiri dari imbuhan awal terulangkan dan imbuhan akhir terulangkan.

Imbuhan awal terulangkan

Dalam format, imbuhan awal terulangkan harus ditempatkan sebelum atau di depan tengara ruas. Pada luaran, imbuhan ini akan tercetak sebelum atau di depan setiap nilai ruas. Jika setelah imbuhan awal terulangkan ditempatkan tanda tambah (+), imbuhan ini akan tercetak di depan setiap nilai ruas kecuali nilai ruas pertama. Di bawah ini adalah contoh penggunaan imbuhan awal terulangkan tanpa dan dengan tanda tambah serta luarannya:

PENULIS PERSONAL (300): Salim, Agus %Ratulangi, Sam %Hatta, Muhammad

Format: | ; |v300

Luaran: ; Salim, Agus ; Ratulangi, Sam ; Hatta, Muhammad

Format: | ; |+v300

Luaran: Salim, Agus ; Ratulangi, Sam ; Hatta, Muhammad

Format: | Penulis: |v300

• Luaran: Penulis: Salim, Agus Penulis:
Ratulangi, Sam Penulis: Hatta,
Muhammad

Penempatan tanda tambah setelah imbuhan dalam format kedua di atas menghindarkan timbulnya imbuhan (;) di depan nilai ruas pertama.

Imbuhan awal terulangkan ditempatkan dalam format seolah-olah bagian dari nilai ruas yang akan dicetak. Jika tengara ruas disertai perintah indensi (f,c), maka imbuhan awal terulangkan akan tercetak pada indensi tersebut, seperti terlihat dalam contoh di bawah ini (satu titik digunakan untuk menyatakan satu spasi).

Format: | Penulis: |v300(0,10)
Luaran: .Penulis: Salim, Agus
.....Penulis: Ratulangi,
.....Sam Penulis: Hatta,
.....Muhammad

Imbuhan akhir terulangkan

Dalam format, imbuhan akhir terulangkan harus ditempatkan setelah atau di belakang tengara ruas. Dalam luaran, imbuhan ini akan tercetak setelah setiap nilai ruas. Jika sebelum imbuhan akhir terulangkan ditempatkan tanda tambah (+), imbuhan ini akan tercetak setelah setiap nilai ruas kecuali nilai ruas terakhir.

Di bawah ini adalah contoh penggunaan imbuhan akhir terulangkan tanpa dan dengan tanda tambah serta luarannya:

PENULIS PERSONAL (300): Salim, Agus%Ratulangi,
Sam%Hatta, Muhammad

Format 1: v300| ; |
Luaran: Salim, Agus ; Ratulangi, Sam
; Hatta, Muhammad ;

Format 2: v300+| ; |
Luaran: Salim, Agus ; Ratulangi, Sam
; Hatta, Muhammad

Format 3: v300,| ; |
Luaran: Salim, AgusRatulangi,
SamHatta, Muhammad

Format 4: | (|v300|)
Luaran: (Salim, Agus) (Ratulangi,
Sam) (Hatta, Muhammad)

Penempatan tanda tambah di depan imbuhan dalam format kedua di atas menghindarkan timbulnya imbuhan (;) setelah nilai ruas terakhir. Penempatan koma antara tengara ruas dan imbuhan akhir dalam format 3 menyebabkan imbuhan tidak tercetak pada luaran. Penggunaan imbuhan awal dan akhir terlihat pada format 4.

Imbuhan akhir terulangkan tercetak mengikuti perintah indensi (f,c) untuk ruas yang bersangkutan seperti terlihat pada contoh di bawah ini:

Format 5: "Penulis: "/v300(9,9)+| ; |
Luaran: Penulis:
.....Salim, Agus ;
.....Ratulangi, Sam
.....; Hatta, Muhammad

PENULISAN LEBIH DARI SATU IMBUHAN

Dua imbuhan atau lebih dapat diberikan kepada satu ruas. Ketentuan dan urutan penempatan dua imbuhan atau lebih dalam format adalah sebagai berikut:

Imbuhan awal

Jika lebih dari satu imbuhan awal bersyarat harus digunakan, imbuhan-imbuhan ini harus ditempatkan berurutan dalam format. Setelah imbuhan awal bersyarat terakhir, dapat ditempatkan perintah spasi (vertikal dan horizontal), perintah model luaran (misalnya mhu). Semua perintah yang terletak antara imbuhan awal bersyarat pertama dan tengara ruas akan tercetak pada luaran hanya jika dalam ruas terdapat nilai.

Dengan format di bawah ini, imbuhan awal bersyarat (SUBYEK:), perintah mencetak huruf besar (mhu) dan garis miring (/) akan tercetak pada luaran jika dalam Ruas SUBYEK terdapat nilai:

SUBYEK (620): Bank dan perbankan%Industri
konstruksi

Format: "SUBYEK: ",mhu,| / |+v620
Luaran: SUBYEK: BANK DAN PERBANKAN /
INDUSTRI KONSTRUKSI

Untuk suatu ruas hanya dapat diberikan satu imbuhan awal terulangkan. Imbuhan ini harus ditempatkan tepat sebelum tengara ruas. Imbuhan awal akan

tercetak dengan huruf besar jika ditempatkan setelah perintah model luaran **mdu** atau **mhu**.

Imbuhan akhir

Untuk suatu ruas hanya dapat diberikan satu imbuhan akhir terulang. Imbuhan ini harus ditempatkan tepat setelah tengara ruas.

Untuk suatu ruas hanya dapat diberikan satu imbuhan akhir bersyarat. Imbuhan ini harus ditempatkan tepat setelah tengara ruas atau setelah imbuhan akhir terulang.

Imbuhan akhir sebaiknya tidak dipisahkan oleh koma (,). Koma juga sebaiknya tidak ditempatkan di antara tengara ruas dan imbuhan akhir yang pertama karena koma membatalkan imbuhan ini.

Imbuhan akhir akan tercetak dengan huruf besar jika ditempatkan setelah perintah model luaran **mdu** atau **mhu**.

TENGARA RUAS SAMARAN

Imbuhan-imbuhan yang telah dibahas dalam seksi sebelumnya terbatas kegunaannya. Kadang-kadang, dua imbuhan harus diperlihatkan dengan syarat berikut: imbuhan pertama diperlukan jika suatu ruas tidak mempunyai nilai dan imbuhan kedua diperlukan jika ruas tersebut mempunyai nilai. Dalam format, imbuhan ini harus diletakkan di depan atau sebelum tengara ruas samaran.

Untuk format CDS/ISIS terdapat dua tengara ruas samaran: **Dt** (atau **Dt^x**) dan **Nt** (atau **Nt^x**), di mana

D atau **d** merupakan perintah untuk mencetak imbuhan jika dalam ruas terdapat nilai

N atau **n** merupakan perintah untuk mencetak imbuhan jika dalam ruas tidak terdapat nilai

t adalah tengara ruas

^x adalah penanda subruas.

Dalam format, tengara ruas samaran tidak perlu didahului oleh huruf **v** dan hanya dapat digunakan dengan imbuhan bersyarat, misalnya "Penulis tidak diketahui"**n300**. Dalam contoh di bawah ini, imbuhan 'PENULIS: TIDAK DIKETAHUI' tercetak pada luaran jika dalam Ruas PENULIS PERSONAL tidak terdapat nilai; jika ada, yang harus tercetak adalah imbuhan 'PENULIS:' dan nilai ruas.

Cantuman 1:

JUDUL DAN KEPENGARANGAN (200): **^a**Metode statistik

PENULIS PERSONAL (300): Nasution, Andi Hakim

Cantuman 2:

JUDUL DAN KEPENGARANGAN (200): **^a**Pengantar statistik

PENULIS PERSONAL (300): [kosong]

Format: "PENULIS: TIDAK DIKETAHUI"**n300**,
"PENULIS: "**v300**/" JUDUL:
"**v200^a**/#

Luaran: PENULIS: Nasution, Andi Hakim
JUDUL: Metode statistik

PENULIS: TIDAK DIKETAHUI
JUDUL: Pengantar statistik

PERINTAH MENCETAK MFN

Untuk mencetak MFN (*master file number*) atau nomor cantuman induk, diperlukan format berikut: **MFN** atau **MFN(d)**. **d** adalah jumlah angka yang harus digunakan untuk membentuk MFN. Jika **d** tidak diisi, maka jumlah angka dalam nomor adalah enam. Dalam format, perintah dapat ditulis dengan huruf besar (MFN) atau huruf kecil (mf). Nomor cantuman pada umumnya diperlukan pada waktu mencetak cantuman baru untuk diperiksa karena cara yang termudah untuk membuka cantuman dalam lembar kerja ketik adalah dengan menggunakan MFN. Di bawah ini adalah contoh format untuk mendapatkan MFN dan masing-masing luarannya:

MFN	000005
MFN(4)	0005

Untuk menghilangkan rangkaian nol di depan, harus digunakan format berikut: **F(MFN,1,0)**.

PERINTAH MODEL LUARAN

Perintah model luaran memungkinkan luaran dicetak dengan huruf besar, dalam bentuk sebagaimana data diketik, atau dalam bentuk sebagaimana seharusnya cantuman tampak dalam katalog atau indeks. Model luaran dinyatakan dalam format dengan tiga huruf: **MMC** (atau ditulis dengan huruf kecil

mnc), di mana:

M (yang pertama) adalah huruf yang harus ada dalam kombinasi tersebut.

M (yang kedua; singkatan dari kata Inggris *mode*) adalah kode berupa huruf yang menyatakan model luaran. Kode untuk model luaran adalah **P**, **H** dan **D**. Fungsi ketiga huruf ini dijelaskan di bawah ini:

P (singkatan dari kata Inggris *proof*) adalah kode untuk model luaran kasar.

Dengan kode ini, data akan tercetak sebagaimana data diketik dan tanda < >, %, dan simbol subruas (^) tetap tampak. Kode ini digunakan untuk mencetak cantuman yang baru diketik dan harus diperiksa.

H (singkatan dari kata Inggris *heading*) adalah kode untuk model tajuk urut. Model luaran ini digunakan jika nilai ruas dipakai sebagai tajuk urut. Tajuk urut akan dibahas dalam artikel lain.

Dengan kode ini, data akan tercetak tanpa tanda < >, %, dan penanda subruas (^). Pada luaran, penanda subruas ^a akan diganti dengan titik koma (;); penanda subruas ^b hingga ^i akan diganti dengan koma (,); penanda subruas lainnya akan diganti dengan titik (.).

D (singkatan dari kata Inggris *data*) adalah kode untuk model data.

Kode ini menghasilkan luaran yang hampir sama dengan luaran yang dihasilkan oleh kode H, kecuali dalam hal-hal berikut. Pada luaran, setiap nilai ruas akan diakhiri dengan titik (.) dan dua spasi. Jika nilai ruas telah diakhiri dengan titik pada waktu diketik, maka setiap nilai ruas hanya akan diikuti oleh dua spasi.

C (singkatan dari kata Inggris *case*) adalah kode berupa huruf untuk mengubah huruf teks.

Kode untuk huruf teks adalah **U** (untuk mengubah huruf teks menjadi huruf besar) dan **L** (agar huruf teks sama seperti pada waktu cantuman diketik).

Di bawah ini adalah contoh penggunaan kode model luaran dalam format dan luaran yang dihasilkan. Pada luaran dari format 1, luaran tercetak seperti pada waktu nilai ruas diketik (penanda subruas

(^a) tetap tercetak dan kata 'The' tetap dalam tanda kurung segitiga). Pada luaran dari format 2, kata 'The' tidak diapit oleh kurung segitiga. Dengan format 3, luaran diakhiri oleh titik. Pada luaran dari format 4, judul tercetak dengan huruf besar dan diakhiri dengan titik dan dua spasi. Dua spasi ini hanya tampak jika judul diikuti oleh nilai ruas lain.

JUDUL DAN KEPENGARANGAN (200): ^a<The>
Psychology of everyday things^bby Donald
A. Norman

Format 1: mp1, v200

Luaran: ^a<The> Psychology of everyday things^bby Donald A. Norman

Format 2: mh1, v200^a

Luaran: The Psychology of everyday things

Format 3: md1, v200^a

Luaran: The Psychology of everyday things.

Format 4: mdu, v200^a

Luaran: THE PSYCHOLOGY OF EVERYDAY THINGS.

Penggunaan model luaran pada tiga subruas tampak pada contoh di bawah ini. Pada luaran dari format 6, penanda subruas ^b dan ^c digantikan oleh koma. Dengan format 7, nilai ruas tercetak dengan huruf besar dan diakhiri dengan titik dan dua spasi; sedangkan penanda subruas ^b dan ^c digantikan oleh koma. Pada luaran dari format 8, penanda subruas ^b digantikan oleh koma dan nilai ruas diakhiri dengan titik dan dua spasi.

PENERBIT (400): ^aJakarta^bBalai Pustaka^c1995

Format 5: mp1, v400

Luaran: ^aJakarta^bBalai Pustaka ^c1995

Format 6: mh1, v400

Luaran: Jakarta, Balai Pustaka, 1995

Format 7: mdu, v400

Luaran: JAKARTA, BALAI PUSTAKA, 1995.

Format 8: md1, v400

Luaran: Jakarta, Balai Pustaka, 1995.

Jumlah model luaran dalam format tidak dibatasi. Suatu model luaran berlaku untuk semua ruas/subruas yang terletak sesudahnya sampai model luaran lain. Jika model luaran tidak ditentukan, CDS/ISIS akan menggunakan **mpl**.

MENYIMPAN FORMAT CETAK ATAU FORMAT TAMPILAN

Format cetak atau format tampilan harus dibuat di layar dan dapat disimpan untuk digunakan kemudian. Format yang telah tersimpan dapat diubah.

Membuka layar di mana format harus dibuat

Untuk membuka layar di mana format harus dibuat, ikuti prosedur berikut. Prosedur ini juga dapat digunakan untuk membuka kembali format yang pernah dibuat dan disimpan.

- 1: Tekan **D** pada menu utama.
- 2: Tekan **U** pada menu **Data Base Definition Services**.

Pertanyaan berikut kemudian muncul:
Data base name:

- 3: Ketik **bibli**, kemudian tekan **Enter**.
- 4: Tekan **H** pada menu **Data Base Definition Services**.

Pertanyaan berikut kemudian muncul:
Format name:

- 5: Ketik **bibli**, kemudian tekan **Enter**.

Setelah langkah 5 di atas akan tampak layar kosong dengan nama pangkalan data (**BIBLI**) tercetak di kiri atas dan nama format (**BIBLI**) di kanan atas.

Format dapat diketik mulai dari kursor. Tombol-tombol untuk pengetikan cantuman dalam lembar kerja ketik (misalnya **F3**, **F4**, **F5**, **F6**, dan **Insert**) dapat digunakan dalam pengetikan format. Untuk mengakhiri pembuatan format dan menyimpannya, tekan **Enter**. Layar di mana format harus dibuat tampak pada gambar di bawah ini.

Jika Anda telah menciptakan pangkalan data dengan nama lain, ketik nama tersebut (untuk menggantikan kata 'bibli') pada langkah 3. Jika Anda ingin menciptakan format dengan nama lain, ketik nama tersebut (untuk menggantikan kata 'bibli') pada langkah 5.

Untuk melihat daftar format yang pernah dibuat dan tersimpan, tekan **K** pada langkah 4.

Membuka kembali format yang pernah dibuat

Untuk membuka kembali format yang pernah dibuat, ikuti prosedur dalam subseksi sebelumnya (**Membuka layar di mana format harus dibuat**) dan ketik nama format pada langkah 5. Membuka kembali suatu format juga dapat dilakukan dengan mengikuti prosedur berikut:

- 1: Tekan **S** pada menu utama.
- 2: Tekan **F** pada menu **Information Retrieval Services**.

Perintah berikut kemudian muncul:
Please enter/edit format (@xxxxx to use predefined format)

- 3: Ketik **@bibli**, kemudian tekan **Enter**.

Setelah langkah 3 di atas akan tampak sebuah format. Untuk mengakhiri penyuntingan format dan menyimpannya, tekan **F8** lebih dahulu sebelum menekan **Enter**.

Data Base Name: BIBLI

Format name: BIBLI

EDIT: Replace

Layar di mana format harus dibuat

Jika Anda telah membuat format dengan nama lain dan ingin membukanya, ketik nama tersebut dan dahului dengan '@' untuk menggantikan '@bibli' pada langkah 3.

FORMAT CONTOH

Format di bawah ini menghasilkan luaran berupa katalog. Ruas-ruas yang terdapat dalam format ini adalah NOMOR PANGGIL, PENULIS PERSONAL, JUDUL DAN KEPENGARANGAN, EDISI, PENERBIT, DESKRIPSI FISIK, CATATAN, dan SUBYEK. Ketik format ini mulai dari kursor dan jangan menekan **Enter** sebelum seluruh rangkaian terketik:

```
mhl,v610^a,' ',v610^b,' ',v610^c/
v300+| ; | /v200^a(4,0)," /
"v200^b," - "v260,'. - 'v400^a,
' : ',v400^b,', ',v400^c'.'/
v460(4,0)/v500(4,0)%##v620(4,0)+|
; | /###
```

Luaran:

025 Phi m
Philipps, Eva

Membina perpustakaan : pedoman kerja perpustakaan sederhana bidang teknologi tepat guna dan pembangunan desa / Eva Philipps ; alih bahasa, Agus Permadi. - Jakarta : German Appropriate Technology Exchange ; Pusat Dokumentasi dan Informasi Ilmiah, 1992.

177 hal.

Perpustakaan ; Administrasi perpustakaan

Penjelasan mengenai format di atas, bagian demi bagian, adalah sebagai berikut:

mhl adalah kode model luaran. Dengan kode ini, tanda-tanda tertentu (misalnya < >, yang pada umumnya digunakan untuk menempatkan kata sandang Inggris *The*, *A*, dan *An*) dalam Ruas JUDUL DAN KEPENGARANGAN tidak tercetak pada luaran. Huruf luaran tidak berubah (huruf kecil tidak

menjadi huruf besar). Koma setelah **mhl** dan koma-koma lainnya (kecuali koma dalam tanda petik tunggal) digunakan untuk memisahkan perintah dan tidak menimbulkan akibat apapun pada luaran.

v610^a, **'v610^b,** **'v610^c** untuk menampilkan nilai Ruas NOMOR PANGGIL yang terbagi dalam tiga subruas (^a, ^b, dan ^c). Subruas a berisi nomor klas, Subruas b berisi tiga huruf pertama nama keluarga penulis atau judul, dan Subruas c berisi huruf pertama judul selain kata sandang. Imbuhan spasi (' ') digunakan untuk memisahkan nilai Subruas b dari nilai Subruas a, dan nilai Subruas c dari nilai Subruas b.

Jika nomor panggil disusun ke bawah, format harus ditulis sebagai berikut: **v610^a/v610^b/v610^c%#**. **%#** dalam format ini dimaksudkan untuk meng-hapus spasi vertikal (antara nomor panggil dan nama penulis) yang timbul jika Subruas c tidak mempunyai nilai.

/v300+ | ; | untuk menampilkan nilai Ruas PENULIS PERSONAL. Garis miring (/) menempatkan ruas ini pada baris baru. Imbuhan (+ | ; |) memisahkan nama-nama dengan spasi, titik koma, spasi.

/v200^a(4,0) untuk menampilkan nilai Subruas a dari Ruas JUDUL DAN KEPENGARANGAN. Subruas a berisi judul buku. Garis miring (/) menempatkan judul pada baris baru. Perintah indensi ((4,0)) menempatkan baris pertama judul mulai dari posisi 5 dan baris-baris selanjutnya mulai dari margin kiri.

" / "v200^b(4,0), untuk menampilkan nilai Subruas b dari Ruas JUDUL DAN KEPENGARANGAN. Subruas b berisi pernyataan kepengarangan. Imbuhan bersyarat (" / ") digunakan karena pernyataan kepengarangan tidak selalu ada. Jika tidak terdapat pernyataan kepengarangan, imbuhan garis miring tidak tercetak.

" - "v260 untuk menampilkan nilai Ruas EDISI. Dalam buku tidak selalu terdapat keterangan edisi. Imbuhan bersyarat (" - ") tercetak jika terdapat keterangan edisi.

' - ',v400^a,' : ',v400^b,', 'v400^c'. untuk menampilkan tiga komponen dalam Ruas PENERBIT, yaitu kota terbit (Subruas a), nama penerbit (Subruas b) dan tahun terbit (Subruas c). Imbuhan tak bersyarat

(' -') diperlukan karena dalam Ruas PENERBIT selalu terdapat nilai. Imbuhan tidak bersyarat kedua (' : ') memisahkan kota terbit dari nama penerbit. Imbuhan tidak bersyarat ketiga (' , ') memisahkan nama penerbit dari tahun terbit. Imbuhan terakhir (' .') mengakhiri nilai ruas ini dengan titik.

/v460(4,0) untuk menampilkan nilai Ruas DESKRIPSI FISIK pada baris baru mulai dari posisi 5. Baris selanjutnya, jika ada, akan tercetak pada baris baru mulai dari margin kiri.

/v500(4,0) untuk menampilkan nilai Ruas CATATAN pada baris baru mulai dari posisi 5. Baris selanjutnya, jika ada, akan tercetak pada baris baru mulai dari margin kiri.

%## untuk menempatkan nilai Ruas SUBYEK pada baris baru dan menyisipkan satu spasi vertikal di atasnya.

v620(4,0)+ | ; | untuk menampilkan nilai Ruas SUBYEK mulai dari posisi 5. Tajuk-tajuk subyek dipisahkan oleh spasi, titik koma, spasi.

/### untuk menciptakan tiga spasi vertikal antara cantuman yang satu dengan cantuman lainnya.

PERINGATAN KESALAHAN

Kesalahan dapat terjadi dalam format, terutama pada format yang panjang. Pada waktu menjalankan format, CDS/ISIS memeriksa kode demi kode dalam format. Jika terdapat kesalahan dalam format, peringatan dalam bentuk kalimat berikut akan diberikan:

Format error X, PFT=nama

di mana X adalah angka kode peringatan dan nama adalah nama format. Untuk memperbaiki kesalahan dalam format, tekan Y setelah pertanyaan **Edit format (Y/N)?**.

Kalimat peringatan tidak selalu dengan jelas menunjukkan bentuk kesalahan dalam format. Di samping itu, satu kalimat peringatan dapat menunjukkan lebih dari satu jenis kesalahan. Di bawah ini adalah beberapa angka kode yang menunjukkan bentuk kesalahan dalam format yang telah dibahas dalam artikel ini.

- 51 Imbuhan untuk suatu ruas terlalu banyak jumlahnya.
- 54 Kode untuk imbuhan terulangkan tidak lengkap, misalnya penulisan v300+ ; |, yang seharusnya v300+ | ; |.
- 61 Tengara ruas tidak disertai huruf v, misalnya penulisan 200^a, yang seharusnya v200^a.
- 99 Kode model luaran tidak benar, misalnya penulisan mel, yang seharusnya mhl, mdl atau mpl.